Le but du problème est de déterminer l'unique polynôme P, de degrén, qui coincide avec la fonction f en $n+1$ points régulièrement espacés sur l'intervalle $[a,b]$.
On dit que ce polynôme est le polynôme d'interpolation de Lagrange de la fonction.
Donnons d'abord des valeurs à a, b, n pour essayer nos procédures.

```
> a:=0;b:=1;n:=7;
```

$$a := 0$$
$$b := 1$$
$$n := 7$$

Créons maintenant une procédure X de paramètres k,a,b,n qui calcule : $x_k = a + k\dfrac{b-a}{n}$.
Quand k varie de 0 à n, on a ainsi les abcisses des $n+1$ points de contact de f et P.

```
> X:=proc(k,a,b,n)
> evalf(a+k*(b-a)/n)
> end;
```

$$X := \textbf{proc}(k, a, b, n)\ \text{evalf}(a + k * (b - a)/n)\ \textbf{end proc}$$

Exprimons maintenant la solution mathématique du problème posé. On a :

$$P(x) = \sum_{i=0}^{n} f(x_i) \left(\prod_{\substack{j=0 \\ j\neq i}}^{n} \frac{x - x_j}{x_i - x_j} \right)$$

On va créer une procédure *facteur* de paramètres i,a,b,n qui calcule $\displaystyle\prod_{\substack{j=0 \\ j\neq i}}^{n} \frac{x - x_j}{x_i - x_j}$.

```
> facteur:=proc(i,a,b,n)
> local p,j;
> p:=1;
> for j from 0 to n do
> if i<>j then p:=p*(x-X(j,a,b,n))/(X(i,a,b,n)-X(j,a,b,n)) fi
> od;
> collect(p,x)
> end;
```

facteur := **proc**(i, a, b, n)
local p, j;
  $p := 1$;
  for j **from** 0 **to** n **do**
    if $i \neq j$ **then** $p := p * (x - \mathrm{X}(j, a, b, n))/(\mathrm{X}(i, a, b, n) - \mathrm{X}(j, a, b, n))$ **end if**
  end do;
  collect(p, x)
end proc

On essaye maintenant ces deux procédures :

```
> X(2,a,b,n);
```

.2857142857

```
>  facteur(2,a,b,n);
```

$$-3431.429166\,x^7 + 12745.30833\,x^6 - 18907.87499\,x^5 + 14205.91666\,x^4 - 5615.195834\,x^3 + 1076.775000\,x^2 - 73.50000000\,x$$

Maintenant, on va créer une procédure *interpole* de paramètres f,a,b,n qui calcule le polynôme d'interpolation demandé. Dans cette procédure, on calcule la somme demandée au moyen d'une boucle, mais on aurait aussi pu utiliser la fonction `sum`.

```
>  interpole:=proc(f,a,b,n)
>  local s,i;
>  s:=0;
>  for i from 0 to n do s:=s+f(X(i,a,b,n))*facteur(i,a,b,n) od;
>  collect(s,x)
>  end;
```

interpole := **proc**(f, a, b, n)
local s, i;
$s := 0$;
for i **from** 0 **to** n **do** $s := s + f(\mathrm{X}(i, a, b, n)) * \mathrm{facteur}(i, a, b, n)$ **end do**;
collect(s, x)
end proc

Pour pouvoir essayer cette procédure, il nous faut créer une fonction f.

```
>  f:=x->1/(1+x*x);
```

$$f := x \to \frac{1}{1+x^2}$$

Essayons donc :

```
>  interpole(f,a,b,n);
```

$$-.10341825\,x^7 + .7290116\,x^6 - 1.8375295\,x^5 + 1.9341392\,x^4 - .25593455\,x^3 - .964341845\,x^2 - .0019297715\,x + 1.000000000$$

Cela nous donne bien un polynôme du bon degré.
On n'est pas obligés de travailler avec les valeurs des paramètres données au début :

```
>  interpole(f,-1,1,5);
```

$$-.22\,10^{-8}\,x^5 + .3535067878\,x^4 + .23\,10^{-8}\,x^3 - .8484162887\,x^2 - .100\,10^{-9}\,x + .9949095018$$

Le mieux serait de voir ce qui se passe sur un graphe.
Pour cela, on crée une procédure *dessine*, toujours de paramètres f,a,b,n et qui dessine sur un même graphe P et f.

```
>  dessine:=proc(f,a,b,n)
>  plot({f(x),interpole(f,a,b,n)},x=a..b,scaling=constrained)
>  end;
```

dessine := **proc**(f, a, b, n)
plot($\{f(x), \mathrm{interpole}(f, a, b, n)\}, x = a..b, scaling = constrained$)
end proc

Il nous reste à regarder :

```
> dessine(x->1/(1+x**8),0,2,11);
```